The   game of Pig   is a multiplayer game played with a single six-sided die.   The
object of the game is to reach   100   points or more.  
Play is taken in turns.   On each person's turn that person has the option of either:

 Rolling the dice:   where a roll of two to six is added to their score for that turn and the player's turn continues as the player is given the same choice again;   or a roll of   1   loses the player's total points   for that turn   and their turn finishes with play passing to the next player.
 Holding:   the player's score for that round is added to their total and becomes safe from the effects of throwing a   1   (one).   The player's turn finishes with play passing to the next player.


Task

Create a program to score for, and simulate dice throws for, a two-person game.



Related task

   Pig the dice game/Player


void main() {
    import std.stdio, std.string, std.algorithm, std.random;
    enum maxScore = 100;
    enum playerCount = 2;
    immutable confirmations = ["yes", "y", ""];
 
    int[playerCount] safeScore;
    int player, score;
 
    while (true) {
        writef(" Player %d: (%d, %d). Rolling? (y/n) ", player,
               safeScore[player], score);
        if (safeScore[player] + score < maxScore &&
            confirmations.canFind(readln.strip.toLower)) {
            immutable rolled = uniform(1, 7);
            writefln(" Rolled %d", rolled);
            if (rolled == 1) {
                writefln(" Bust! You lose %d but keep %d\n",
                         score, safeScore[player]);
            } else {
                score += rolled;
                continue;
            }
        } else {
            safeScore[player] += score;
            if (safeScore[player] >= maxScore)
                break;
            writefln(" Sticking with %d\n", safeScore[player]);
        }
 
        score = 0;
        player = (player + 1) % playerCount;
    }
 
    writefln("\n\nPlayer %d wins with a score of %d",
             player, safeScore[player]);
}